

6.5.1.1 Loads and Stores

Table 6-1. Load and Store Instructions

Source Form	Operation	Address Mode	Object Code	Cycles	Cyc-by-Cyc Details	Effect on CCR	
						V I 1 H	IN Z C
LDA #opr8i LDA opr8a LDA opr16a LDA oprx16,X LDA oprx8,X LDA ,X LDA oprx16,SP LDA oprx8,SP	Load Accumulator from Memory $A \leftarrow (M)$	IMM DIR EXT IX2 IX1 IX SP2 SP1	A6 ii B6 dd C6 hh ll D6 ee ff E6 ff F6 9E D6 ee ff 9E E6 ff	2 3 4 4 3 3 5 4	pp rpp prpp prpp rpp rfp pprpp prpp	0 1 1 -	- † † -
LDHX #opr16i LDHX opr8a LDHX opr16a LDHX ,X LDHX oprx16,X LDHX oprx8,X LDHX oprx8,SP	Load Index Register (H:X) $H:X \leftarrow (M:M + \$0001)$	IMM DIR EXT IX IX2 IX1 SP1	45 jj kk 55 dd 32 hh ll 9E AE 9E BE ee ff 9E CE ff 9E FE ff	3 4 5 5 6 5 5	ppp rrpp prpp prpp pprrpp prpp prpp	0 1 1 -	- † † -
LDX #opr8i LDX opr8a LDX opr16a LDX oprx16,X LDX oprx8,X LDX ,X LDX oprx16,SP LDX oprx8,SP	Load X (Index Register Low) from Memory $X \leftarrow (M)$	IMM DIR EXT IX2 IX1 IX SP2 SP1	AE ii BE dd CE hh ll DE ee ff EE ff FE 9E DE ee ff 9E EE ff	2 3 4 4 3 3 5 4	pp rpp prpp prpp rpp rfp pprpp prpp	0 1 1 -	- † † -
STA opr8a STA opr16a STA oprx16,X STA oprx8,X STA ,X STA oprx16,SP STA oprx8,SP	Store Accumulator in Memory $M \leftarrow (A)$	DIR EXT IX2 IX1 IX SP2 SP1	B7 dd C7 hh ll D7 ee ff E7 ff F7 9E D7 ee ff 9E E7 ff	3 4 4 3 2 5 4	wpp pwpp pwpp wpp wp ppwpp pwpp	0 1 1 -	- † † -
STHX opr8a STHX opr16a STHX oprx8,SP	Store H:X (Index Reg.) $(M:M + \$0001) \leftarrow (H:X)$	DIR EXT SP1	35 dd 96 hh ll 9E FF ff	4 5 5	wwpp pwpp pwpp	0 1 1 -	- † † -
STX opr8a STX opr16a STX oprx16,X STX oprx8,X STX ,X STX oprx16,SP STX oprx8,SP	Store X (Low 8 Bits of Index Register) in Memory $M \leftarrow (X)$	DIR EXT IX2 IX1 IX SP2 SP1	BF dd CF hh ll DF ee ff EF ff FF 9E DF ee ff 9E EF ff	3 4 4 3 2 5 4	wpp pwpp pwpp wpp wp ppwpp pwpp	0 1 1 -	- † † -

Load A and load X cause an 8-bit value to be read from memory into accumulator A or into the X register. Load H:X causes one 8-bit value to be read from memory into the H register and a second 8-bit value to be read from the next sequential memory location into the X register. Load A and load X each allow eight different addressing modes for maximum flexibility in accessing memory. LDHX allows seven different addressing modes to specify the memory locations of the values being read.

The following instructions demonstrate some of the uses for load instructions. This collection of instructions is not intended to be a meaningful program. Rather, they are unrelated load instructions to demonstrate the many possible addressing modes that allow access to memory in different ways.

```

226      ; load A - various addressing modes
227      ; immediate (IMM) addressing mode examples
228 C089 A6 55      lda    #$55      ;IMM - $ means hexadecimal
229 C08B A6 64      lda    #100      ;decimal 100 (hexadecimal $64)
230 C08D A6 3F      lda    #%00111111 ;% means binary
231 C08F A6 41      lda    #'A'      ;single quotes around ASCII
232 C091 A6 8D      lda    #illegalOp ;label used as immediate value
233      ; direct (DIR) addressing mode examples
234 C093 B6 55      lda    $55      ;load from address $0055
235 C095 B6 9D      lda    directByte ;label as a direct address
236      ; extended (EXT) addressing mode
237 C097 C6 FFFE      lda    $FFFE      ;high byte of reset vector
238 C09A C6 0101      lda    extByte      ;label used as an address
239 C09D C6 C09D      lda    *          ;* means "here", loads opcode
240 C0A0 C6 009D      lda    fwdRef      ;forces ext addressing mode
241      ; not all assemblers treat forward references the same way
242      0000 009D fwdRef: equ    directByte ;forward referenced direct
243
244 C0A3 45 C007      ldhx   #stringBytes ;point at string in flash
245      ; indexed addressing mode (relative to H:X index register pair)
246 C0A6 D6 4081      lda    (moveBlkl-stringBytes),x ;IX2 mode
247 C0A9 E6 01      lda    1,x          ;IX1 - 8-bit offset
248 C0AB F6          lda    ,x          ;IX - no offset
249
250      ; indexed addressing mode (relative to SP stack pointer)
251 C0AC 45 0001      ldhx   #1
252 C0AF 94          txs          ;temp move SP for 16-bit offset ex.
253 C0B0 9ED6 012C      lda    300,sp      ;SP2 - 16-bit offset
254 C0B4 9EE6 01      lda    1,sp        ;SP1 - 8-bit offset

```

Since one operand input to the arithmetic logic unit (ALU) is connected to the A accumulator, you typically need to use an LDA instruction to read one value into A before performing mathematical or logical operations involving a second operand.

```

; add A + B (assumes sum is < or = 255)
      lda    oprA      ;oprA -> accumulator
      add    oprB      ;oprA + oprB -> accumulator

```

In some cases, you can plan your program so that the results that were stored in accumulator A as the result of one operation can be used as an operand in a subsequent operation. This can save the need to store one result and reload the accumulator with the next operand.

```

; add A + B + C (assumes sum is < or = 255)
      lda    oprA      ;oprA -> accumulator
      add    oprB      ;oprA + oprB -> accumulator
      add    oprC      ;accum. + oprC -> accum.

```

The next example shows an intermediate value being saved on the stack. This is sometimes faster than storing temporary results in memory. The amount of savings depends on what addressing mode would be needed to store the temporary value in memory and whether the X register was needed for something else at the time.

```

; compute (A + B) - (C + D) (assumes no carry or borrow)
      lda    oprC      ;oprC -> accumulator
      add    oprD      ;oprC + oprD -> accumulator
      psha          ;intermediate result to SP+1
      lda    oprA      ;oprA -> accumulator
      add    oprB      ;oprA + oprB -> accumulator
      sub    1,sp      ;(A+B)-(C+D) to accumulator
      ais    #1        ;deallocate local space

```

Table 6-2. BSET, BCLR, Move, and Transfer Instructions

Source Form	Operation	Address Mode	Object Code	Cycles	Cyc-by-Cyc Details	Effect on CCR			
						V	I	H	I N Z C
BSET $n, opr8a$	Set Bit n in Memory ($M_n \leftarrow 1$)	DIR (b0)	10 dd	5	rfwpp				
		DIR (b1)	12 dd	5	rfwpp				
		DIR (b2)	14 dd	5	rfwpp				
		DIR (b3)	16 dd	5	rfwpp				
		DIR (b4)	18 dd	5	rfwpp	-	1	1	- - - -
		DIR (b5)	1A dd	5	rfwpp				
		DIR (b6)	1C dd	5	rfwpp				
		DIR (b7)	1E dd	5	rfwpp				
BCLR $n, opr8a$	Clear Bit n in Memory ($M_n \leftarrow 0$)	DIR (b0)	11 dd	5	rfwpp				
		DIR (b1)	13 dd	5	rfwpp				
		DIR (b2)	15 dd	5	rfwpp				
		DIR (b3)	17 dd	5	rfwpp				
		DIR (b4)	19 dd	5	rfwpp	-	1	1	- - - -
		DIR (b5)	1B dd	5	rfwpp				
		DIR (b6)	1D dd	5	rfwpp				
		DIR (b7)	1F dd	5	rfwpp				
MOV $opr8a, opr8a$	Move	DIR/DIR	4E dd dd	5	rpwpp				
MOV $opr8a, X+$	$(M)_{destination} \leftarrow (M)_{source}$	DIR/IX+	5E dd	5	rfwpp	0	1	1	- - 1 -
MOV $\#opr8i, opr8a$	In IX+/DIR and DIR/IX+ Modes,	IMM/DIR	6E ii dd	4	pwpp				
MOV $,X+, opr8a$	$H:X \leftarrow (H:X) + \$0001$	IX+/DIR	7E dd	5	rfwpp				
TAX	Transfer Accumulator to X (Index Register Low) $X \leftarrow (A)$	INH	97	1	p	-	1	1	- - - -
TXA	Transfer X (Index Reg. Low) to Accumulator $A \leftarrow (X)$	INH	9F	1	p	-	1	1	- - - -
TAP	Transfer Accumulator to CCR $CCR \leftarrow (A)$	INH	84	1	p	1	1	1	1 1 1 1
TPA	Transfer CCR to Accumulator $A \leftarrow (CCR)$	INH	85	1	p	-	1	1	- - - -
NSA	Nibble Swap Accumulator $A \leftarrow (A[3:0]:A[7:4])$	INH	62	1	p	-	1	1	- - - -

6.5.1.2 Bit Set and Bit Clear

Bit set (BSET) and bit clear (BCLR) instructions can be thought of as bit-sized store instructions, but these instructions actually read a full 8-bit location, modify the specified bit, and then re-write the whole 8-bit location. In certain cases, such as when the target location is something other than a RAM variable, this subtle behavior can lead to unexpected results. If a BSET or BCLR instruction attempts to change a bit in a nonvolatile memory location, naturally, the bit will not change because nonvolatile memories require a more complex sequence of operations to make changes.

Some status bits are cleared by a sequence involving a read of the status bit followed by a write to another register in the peripheral module. Some users are surprised to find that a BSET or BCLR instruction has satisfied the requirement to read the status register. To avoid such problems, just remember that the BSET and BCLR instructions are read-modify-write instructions that access a full 8-bit location in parallel.

Some control or I/O registers do not access the same physical logic states for reads and writes. In general, do not use read-modify-write instructions on these locations because they may produce unexpected results.

276				; BSET example - turns on TE without changing RE
277	C0D3	16	1B	bset TE,SCI1C2 ;enable SCI transmitter
278				; functionally equivalent to...
279	C0D5	B6	1B	lda SCI1C2 ;read current SCCR2 value
280	C0D7	AA	08	ora #mTE ;OR in TE bit (mask)
281	C0D9	B7	1B	sta SCI1C2 ;update value in SCCR2

6.5.1.3 Memory-to-Memory Moves

Move instructions can be helpful in an accumulator architecture like the HCS08 where the number of registers is limited. MOV performs a read of an 8-bit value from one memory location and stores the value in a different location. Like the load and store instructions, MOV causes the N and Z bits in the CCR to be updated according to the value of the data being moved.

Although load and store instructions could be used to do the same thing as a MOV instruction, MOV does not require the accumulator to be saved so that A can be used as the transport means for the move operation. In many cases, the MOV approach is faster and smaller (object code size) than the load-store combination. MOV allows four different address mode combinations to specify the source and destination locations for the move.

The following example shows how move instructions can be used to initialize several register values.

284	C0DB	6E	03	00	mov	#\$03,PTAD	;0011 to 4 LS bits
285	C0DE	6E	0F	03	mov	#\$0F,PTADD	;make 4 LS bits outputs
286	C0E1	6E	F0	01	mov	#\$F0,PTAPE	;pullups on 4 MS bits

The next example shows a string move operation using load and store instructions rather than move instructions.

288					; block move example to move a string to a RAM block
289	C0E4	45	00	88	ldhx #moveBlk1 ;point at destination block
290	C0E7	D6	BF	7F	movLoop1: lda (stringBytes-moveBlk1),x ;get source byte
291	C0EA	27	04		beq dunLoop1 ;null terminator ends loop
292	C0EC	F7			sta ,x ;save to destination block
293	C0ED	5C			incx ;next location (assumes DIR)
294	C0EE	20	F7		bra movLoop1 ;continue loop
295					dunLoop1:

6.5.1.4 Register Transfers and Nibble Swap

TAX and TXA offer an efficient way to transfer a value from A to X or from X to A. Depending on whether the X register is already being used, this can be an efficient way to temporarily save the accumulator value so A can be used for some other operation.

TAP and TPA provide a means for moving the value from A into the CCR (processor status byte) or from the CCR into A. This is used more in development tools like debug monitors than in normal user programs.

The nibble swap A (NSA) instruction exchanges the upper and lower nibbles of the accumulator (A). An 8-bit value is called a byte and a nibble is the upper- or lower-order four bits of a byte. Each nibble

Central Processor Unit (CPU)

corresponds to exactly one hexadecimal digit. This instruction is useful for conversions between binary or hexadecimal and ASCII, and for operations on binary-coded-decimal (BCD) numbers.

```
*****
* chexl - convert upper nibble of A to ASCII
* chexr - convert lower nibble of A to ASCII
* on entry A contains any binary (hexadecimal) number
* returns with resulting ASCII character in A
*****
chexl:      nsa          ;swap nibble into low half
chexr:      and    #$0F    ;strip off upper nibble
            add    #$30    ;now $30 - $3F
            cmp    #$39    ;check for < or = '9'
            bls    dunChex ;if so, just return
            add    #7      ;adjust to $41-$46
dunChex:    rts          ;return with ASCII in A
*****
```

6.5.2 Math Instructions

Math instructions include the traditional add, subtract, multiply, and divide operations, a collection of utility instructions including increment, decrement, clear, negate (two's complement), compare, and test, and a decimal adjust instruction for computations involving BCD numbers. The compare instructions are actually subtract operations where the CCR bits are affected but the result is not written back to a CPU register. The test instructions affect the N and Z condition code bits, but do not affect the tested value.

6.5.2.1 Add, Subtract, Multiply, and Divide

Table 6-3. Add, Subtract, Multiply, and Divide Instructions

Source Form	Operation	Address Mode	Object Code	Cycles	Cyc-by-Cyc Details	Effect on CCR			
						V	I	H	I N Z C
ADC #opr8i ADC opr8a ADC opr16a ADC oprx16,X ADC oprx8,X ADC ,X ADC oprx16,SP ADC oprx8,SP	Add with Carry $A \leftarrow (A) + (M) + (C)$	IMM DIR EXT IX2 IX1 IX SP2 SP1	A9 ii B9 dd C9 hh ll D9 ee ff E9 ff F9 9E D9 ee ff 9E E9 ff	2 3 4 4 3 3 5 4	pp rpp prpp prpp rpp rpf pprpp prpp				† 1 1 † - † † †
ADD #opr8i ADD opr8a ADD opr16a ADD oprx16,X ADD oprx8,X ADD ,X ADD oprx16,SP ADD oprx8,SP	Add without Carry $A \leftarrow (A) + (M)$	IMM DIR EXT IX2 IX1 IX SP2 SP1	AB ii BB dd CB hh ll DB ee ff EB ff FB 9E DB ee ff 9E EB ff	2 3 4 4 3 3 5 4	pp rpp prpp prpp rpp rpf pprpp prpp				† 1 1 † - † † †
AIS #opr8i	Add Immediate Value (Signed) to Stack Pointer $SP \leftarrow (SP) + (M)$	IMM	A7 ii	2	pp	-	1	1	- - - -

Table 6-3. Add, Subtract, Multiply, and Divide Instructions (Continued)

Source Form	Operation	Address Mode	Object Code	Cycles	Cyc-by-Cyc Details	Effect on CCR	
						V 1 1 H	I N Z C
AIX #opr8i	Add Immediate Value (Signed) to Index Register (H:X) $H:X \leftarrow (H:X) + (M)$	IMM	AF ii	2	pp	- 1 1 -	- - - -
SUB #opr8i SUB opr8a SUB opr16a SUB oprx16,X SUB oprx8,X SUB ,X SUB oprx16,SP SUB oprx8,SP	Subtract $A \leftarrow (A) - (M)$	IMM DIR EXT IX2 IX1 IX SP2 SP1	A0 ii B0 dd C0 hh ll D0 ee ff E0 ff F0 9E D0 ee ff 9E E0 ff	2 3 4 4 3 3 5 4	pp rpp prpp prpp rpp rfp pprpp prpp	† 1 1 -	- † † †
SBC #opr8i SBC opr8a SBC opr16a SBC oprx16,X SBC oprx8,X SBC ,X SBC oprx16,SP SBC oprx8,SP	Subtract with Carry $A \leftarrow (A) - (M) - (C)$	IMM DIR EXT IX2 IX1 IX SP2 SP1	A2 ii B2 dd C2 hh ll D2 ee ff E2 ff F2 9E D2 ee ff 9E E2 ff	2 3 4 4 3 3 5 4	pp rpp prpp prpp rpp rfp pprpp prpp	† 1 1 -	- † † †
MUL	Unsigned multiply $X:A \leftarrow (X) \times (A)$	INH	42	5	ffffp	- 1 1 0	- - - 0
DIV	Divide $A \leftarrow (H:A) \div (X); H \leftarrow \text{Remainder}$	INH	52	6	fffffp	- 1 1 -	- - † †

The ADD instructions add the value in A to a memory operand and store the result in A. ADC adds the value in A, plus the carry bit from a previous operation, to a memory operand and stores the result in A. This operation allows performance of multibyte additions as demonstrated by the following example.

```

; add 8-bit operand to 24-bit sum
    lda    oprA        ;8-bit operand to A
    add    sum24+2      ;LS byte of 24-bit sum
    sta    sum24+2      ;update LS byte
    lda    sum24+1      ;middle byte of 24-bit sum
    adc    #0           ;propagate any carry
    sta    sum24+1      ;update middle byte
    lda    sum24         ;get MS byte of 24-bit sum
    adc    #0           ;propagate carry into MS byte
    sta    sum24         ;update MS byte

```

The AIX instruction adds a signed 8-bit value to the 16-bit H:X index register pair and stores the result back into H:X. Unlike other arithmetic instructions, AIX does not affect the CCR bits.

```

    ldhx   #tblOfStruct ;H:X pointing at first struct
; aix to update pointer into table of 5-byte structures
    aix    #5           ;point to next 5-byte struct

```

Central Processor Unit (CPU)

The SUB instructions subtract a memory operand from the value in A and store the result in A. The carry status bit acts as a borrow indicator for this subtraction. SBC subtracts a memory operand and the carry bit from a previous operation from the value in A and stores the result back in A. This operation allows performance of multibyte subtractions as demonstrated by the following example.

```
; 16-bit subtract... result16 = oprE - oprF
    lda    oprE+1      ;low half of oprE
    sub    oprF+1      ;oprE(lo) - oprF(lo)
    sta    result16+1  ;low half of result
    lda    oprE         ;high half of oprE
    sbc    oprF         ;oprE(hi) - oprF(hi) - borrow
    sta    result16     ;high half of result
```

MUL multiplies the unsigned 8-bit value in X by the unsigned 8-bit value in A and stores the 16-bit result in X:A where the upper eight bits of the result are stored in X and the lower eight bits of the result are in A. There is no possibility of a carry (or overflow) since the result will always fit into X:A, so C is cleared after this operation.

DIV divides the 16-bit unsigned value in H:A by the 8-bit unsigned value in X and stores the 8-bit result in A and the 8-bit remainder in H. The divisor in X is left unchanged so it could be used in later calculations. Z indicates whether the result was zero, and C indicates whether there was an attempt to divide by zero or if there was an overflow. An overflow will occur if the result was greater than 255.

This first divide example shows a simple 8-bit by 8-bit integer divide to get an 8-bit result.

```
; divide examples
; 8/8 integer divide... A = A/X
    clrh                    ;clear MS byte of dividend
    lda    divid8           ;load 8-bit dividend
    ldx    divisor          ;load divisor
    div                    ;H:A/X -> A, remainder -> H
    sta    quotient8        ;save result
```

The second divide example demonstrates how to use DIV to perform an 8-bit by 8-bit divide and another DIV to resolve the remainder into a fractional result (eight more places to the right of the radix point).

```
; 8/8 integer divide, resolve remainder to 8 fractional bits...
; r8.f8 = A/X, remainder resolved into 8-bit binary fraction
; 16-bit result -> (8-bit integer result).(8-bit fraction)
    clrh                    ;clear MS byte of dividend
    lda    divid8           ;load 8-bit dividend
    ldx    divisor          ;load divisor
    div                    ;H:A/X -> A, remainder -> H
    sta    quotient16       ;upper integer part of result
    clra                    ;H:A = remainder:0
    div                    ;H:A/X -> A
    sta    quotient16+1     ;lower fractional part
```

In the third divide example, we divide an 8-bit dividend by a larger 8-bit divisor to get a 16-bit fractional result where the radix point is just left of the MSB of the result. In a binary fraction, the MSB has a weight of one-half, the next bit to the right has a weight of one-fourth, and so on.

```
; 8/8 fractional divide, 16-bit fractional result
; .r16 = H/X, result is a 16-bit binary fraction
; radix assumed to be in same position for H and X
; 16-bit result -> .(16-bit fraction)
; divid8 and divisor defined so H & X both loaded with one ldhx
      clra          ;clear LS byte of dividend
      ldhx   divid8 ;H:X = dividend:divisor
      div           ;H:A/X -> A, remainder -> H
      sta   quotient16 ;upper byte of result
      clra          ;H:A = remainder:0
      div           ;H:A/X -> A
      sta   quotient16+1 ;next 8 bits of result
```

The fourth divide example uses a technique like long division to do an unbounded 16-bit by 8-bit integer divide.

```
; unbounded 16/8 integer divide (equivalent to long division)
; r16.f8 = H:A/X, result is 16-bit int.8-bit binary fraction
      clrh          ;clear MS byte of dividend
      lda   divid16 ;upper byte of dividend
      ldx   divisor  ;load divisor
      div           ;H:A/X -> A, remainder -> H
      sta   quotient24 ;upper byte of result
      lda   divid16+1 ;H:A = remainder:dividend(lo)
      div           ;H:A/X -> A, remainder -> H
      sta   quotient24+1 ;next byte of result
      clra          ;H:A = remainder:0
      div           ;H:A/X -> A
      sta   quotient24+2 ;fractional bits of result
```

The fifth divide example demonstrates a 16-bit by 8-bit divide with overflow checking.

```
; bounded 16/8 integer divide (with overflow checking)
; r8 = H:A/X, result is 8-bit integer
      ldhx   divid16 ;H:X = 16-bit dividend
      txa          ;H:A = 16-bit dividend
      ldx   divisor  ;X = 8-bit divisor
      div           ;H:A/X -> A, remainder -> H
      bcs   divOvrflow ;Overflow?
      sta   quotient8 ;upper byte of result

divOvrflow:          ;here on overflow
```

Table 6-4. Other Arithmetic Instructions

Source Form	Operation	Address Mode	Object Code	Cycles	Cyc-by-Cyc Details	Effect on CCR	
						V I 1 H	I N Z C
INC <i>opr8a</i> INCA INCX INC <i>opr8,X</i> INC <i>,X</i> INC <i>opr8,SP</i>	Increment $M \leftarrow (M) + \$01$ $A \leftarrow (A) + \$01$ $X \leftarrow (X) + \$01$ $M \leftarrow (M) + \$01$ $M \leftarrow (M) + \$01$ $M \leftarrow (M) + \$01$	DIR INH INH IX1 IX SP1	3C dd 4C 5C 6C ff 7C 9E 6C ff	5 1 1 5 4 6	rfwpp p p rfwpp rfwp prfwpp	1 1 1 1 1 1	– – – – – –
DEC <i>opr8a</i> DECA DECX DEC <i>opr8,X</i> DEC <i>,X</i> DEC <i>opr8,SP</i>	Decrement $M \leftarrow (M) - \$01$ $A \leftarrow (A) - \$01$ $X \leftarrow (X) - \$01$ $M \leftarrow (M) - \$01$ $M \leftarrow (M) - \$01$ $M \leftarrow (M) - \$01$	DIR INH INH IX1 IX SP1	3A dd 4A 5A 6A ff 7A 9E 6A ff	5 1 1 5 4 6	rfwpp p p rfwpp rfwp prfwpp	1 1 1 1 1 1	– – – – – –
CLR <i>opr8a</i> CLRA CLR X CLR H CLR <i>opr8,X</i> CLR <i>,X</i> CLR <i>opr8,SP</i>	Clear $M \leftarrow \$00$ $A \leftarrow \$00$ $X \leftarrow \$00$ $H \leftarrow \$00$ $M \leftarrow \$00$ $M \leftarrow \$00$ $M \leftarrow \$00$	DIR INH INH INH IX1 IX SP1	3F dd 4F 5F 8C 6F ff 7F 9E 6F ff	5 1 1 1 5 4 6	rfwpp p p p rfwpp rfwp prfwpp	0 1 1 1 1 1 1	– – – – – – –
NEG <i>opr8a</i> NEGA NEG X NEG <i>opr8,X</i> NEG <i>,X</i> NEG <i>opr8,SP</i>	Negate $M \leftarrow -(M) = \$00 - (M)$ (Two's Complement) $A \leftarrow -(A) = \$00 - (A)$ $X \leftarrow -(X) = \$00 - (X)$ $M \leftarrow -(M) = \$00 - (M)$ $M \leftarrow -(M) = \$00 - (M)$ $M \leftarrow -(M) = \$00 - (M)$	DIR INH INH IX1 IX SP1	30 dd 40 50 60 ff 70 9E 60 ff	5 1 1 5 4 6	rfwpp p p rfwpp rfwp prfwpp	1 1 1 1 1 1	– – – – – –
CMP <i>#opr8i</i> CMP <i>opr8a</i> CMP <i>opr16a</i> CMP <i>opr16,X</i> CMP <i>opr8,X</i> CMP <i>,X</i> CMP <i>opr16,SP</i> CMP <i>opr8,SP</i>	Compare Accumulator with Memory $A - M$ (CCR Updated But Operands Not Changed)	IMM DIR EXT IX2 IX1 IX SP2 SP1	A1 ii B1 dd C1 hh ll D1 ee ff E1 ff F1 9E D1 ee ff 9E E1 ff	2 3 4 4 3 3 5 4	pp rpp prpp prpp rpp rfp pprpp prpp	1 1 1 1 1 1 1 1	– – – – – – – –
CPHX <i>opr16a</i> CPHX <i>#opr16i</i> CPHX <i>opr8a</i> CPHX <i>opr8,SP</i>	Compare Index Register (H:X) with Memory $(H:X) - (M:M + \$0001)$ (CCR Updated But Operands Not Changed)	EXT IMM DIR SP1	3E hh ll 65 jj kk 75 dd 9E F3 ff	6 3 5 6	prfpp ppp rrfpp prfpp	1 1 1 1	– – – –
CPX <i>#opr8i</i> CPX <i>opr8a</i> CPX <i>opr16a</i> CPX <i>opr16,X</i> CPX <i>opr8,X</i> CPX <i>,X</i> CPX <i>opr16,SP</i> CPX <i>opr8,SP</i>	Compare X (Index Register Low) with Memory $X - M$ (CCR Updated But Operands Not Changed)	IMM DIR EXT IX2 IX1 IX SP2 SP1	A3 ii B3 dd C3 hh ll D3 ee ff E3 ff F3 9E D3 ee ff 9E E3 ff	2 3 4 4 3 3 5 4	pp rpp prpp prpp rpp rfp pprpp prpp	1 1 1 1 1 1 1 1	– – – – – – – –
TST <i>opr8a</i> TSTA TSTX TST <i>opr8,X</i> TST <i>,X</i> TST <i>opr8,SP</i>	Test for Negative or Zero $(M) - \$00$ $(A) - \$00$ $(X) - \$00$ $(M) - \$00$ $(M) - \$00$ $(M) - \$00$	DIR INH INH IX1 IX SP1	3D dd 4D 5D 6D ff 7D 9E 6D ff	4 1 1 4 3 5	rfpp p p rfpp rfp prfpp	0 1 1 1 1 1	– – – – – –
DAA	Decimal Adjust Accumulator After ADD or ADC of BCD Values	INH	72	1	p	U 1 1 1	– 1 1 1

6.5.2.2 Increment, Decrement, Clear, and Negate

Increment and decrement instructions let you adjust the value in A, X, or a memory location by one. Clear instructions let you force an 8-bit value in A, X, H, or a memory location to zero.

Negate instructions perform a two's complement operation that is equivalent to multiplying a signed 8-bit value by negative one. Functionally, this instruction inverts all the bits in A, X, or the memory location and then adds one. The value \$80 represents the signed number -128. The negative of this value would be +128, but the largest positive number that can be represented with a two's complement, 8-bit number is +127. If A was \$80 and you execute a NEGA instruction, the CPU first inverts all the bits to get \$7F and then adds one to get \$80. Since this causes the sign to change from positive to negative, the V bit in the CCR is set to indicate the error.

6.5.2.3 Compare and Test

CMP instructions affect CCR bits exactly like the corresponding SUB instruction, but the result is not stored back into the accumulator so A and the memory operand are left unchanged. Compare instructions compare the contents of A, X, or the H:X register pair to a memory operand. In the case of CPHX, M is the address of the referenced memory location, H corresponds to memory location M, and X corresponds to memory location M+1. CPHX performs a 16-bit subtraction (without storing the result back to H:X).

The test instructions are equivalent to subtracting zero from A, X, or a memory operand. This operation clears V and sets or clears N and Z according to what was in the tested value. The tested value is not changed.

6.5.2.4 BCD Arithmetic

In a binary coded decimal (BCD) number, one hexadecimal digit (4 binary bits) represents a single decimal number from 0 to 9. When two 8-bit BCD numbers are added, the CPU actually does a normal binary addition. Depending on the BCD values involved, this could result in a value that is no longer a valid 2-digit BCD number. Based on the H and C condition code bits that resulted from an ADD or ADC instruction involving two legal BCD numbers, the decimal adjust A (DAA) instruction "corrects" the result to the proper BCD result and sets or clears the C bit as needed to reflect the result of the BCD addition. In the past, this was done with a relatively complex set of instructions that tested the values of each BCD digit of the result and the H and C bits. The DAA instruction greatly simplifies this operation.

The following examples demonstrate two of the possible cases that can result from adding 8-bit BCD numbers and the actions taken by a DAA instruction to correct the results to the appropriate BCD result and carry flag. The first example shows a BCD addition that does not require adjustment. The second example shows a case where the result was not a legal BCD value and the carry did not reflect the correct BCD result. In this second example, the DAA instruction adds a correction factor and adjusts the carry flag to reflect the correct BCD result.

lda	#\$11	;BCD 11
add	#\$22	;11 + 22 = 33
daa		;no adjustment in this case

LDA	#\$59	;BCD 59
ADD	#\$57	;59 + 57 = \$B0
; C=0, H=1, A=\$B0 - wanted 59 + 57 = 116 or A=\$16 with carry set		
DAA		;adds \$66 and sets carry
; \$B0 + \$66 = \$16 with carry bit set		

6.5.3 Logical Operation Instructions

These instructions perform eight bitwise Boolean operations in parallel. For the complement instruction, each bit of the register or memory operand is inverted. The other logical instructions involve two operands, one in the accumulator (A) and the other in memory. Immediate, direct, extended, or indexed (relative to H:X or SP) addressing modes may be used to access the memory operand. Each bit of the accumulator is ANDed, ORed, or exclusive-ORed with the corresponding bit of the memory operand. The result of the logical operation is stored into the accumulator, overwriting the original operand.

Table 6-5. Logical Operation Instructions

Source Form	Operation	Address Mode	Object Code	Cycles	Cyc-by-Cyc Details	Effect on CCR					
						V	I	H	N	Z	C
AND #opr8i AND opr8a AND opr16a AND oprx16,X AND oprx8,X AND ,X AND oprx16,SP AND oprx8,SP	Logical AND $A \leftarrow (A) \& (M)$	IMM DIR EXT IX2 IX1 IX SP2 SP1	A4 ii B4 dd C4 hh ll D4 ee ff E4 ff F4 9E D4 ee ff 9E E4 ff	2 3 4 4 3 3 5 4	pp rpp prpp prpp rpp rfp pprpp prpp	0	1	1	-	-	-
ORA #opr8i ORA opr8a ORA opr16a ORA oprx16,X ORA oprx8,X ORA ,X ORA oprx16,SP ORA oprx8,SP	Inclusive OR Accumulator and Memory $A \leftarrow (A) (M)$	IMM DIR EXT IX2 IX1 IX SP2 SP1	AA ii BA dd CA hh ll DA ee ff EA ff FA 9E DA ee ff 9E EA ff	2 3 4 4 3 3 5 4	pp rpp prpp prpp rpp rfp pprpp prpp	0	1	1	-	-	-
EOR #opr8i EOR opr8a EOR opr16a EOR oprx16,X EOR oprx8,X EOR ,X EOR oprx16,SP EOR oprx8,SP	Exclusive OR Memory with Accumulator $A \leftarrow (A) \oplus (M)$	IMM DIR EXT IX2 IX1 IX SP2 SP1	A8 ii B8 dd C8 hh ll D8 ee ff E8 ff F8 9E D8 ee ff 9E E8 ff	2 3 4 4 3 3 5 4	pp rpp prpp prpp rpp rfp pprpp prpp	0	1	1	-	-	-
COM opr8a COMA COMX COM oprx8,X COM ,X COM oprx8,SP	Complement (One's Complement) $M \leftarrow (\overline{M}) = \$FF - (M)$ $A \leftarrow (\overline{A}) = \$FF - (A)$ $X \leftarrow (\overline{X}) = \$FF - (X)$ $M \leftarrow (\overline{M}) = \$FF - (M)$ $M \leftarrow (\overline{M}) = \$FF - (M)$ $M \leftarrow (\overline{M}) = \$FF - (M)$	DIR INH INH IX1 IX SP1	33 dd 43 53 63 ff 73 9E 63 ff	5 1 1 5 4 6	rfwpp p p rfwpp rfwp prfwpp	0	1	1	-	-	1
BIT #opr8i BIT opr8a BIT opr16a BIT oprx16,X BIT oprx8,X BIT ,X BIT oprx16,SP BIT oprx8,SP	Bit Test (A) & (M) (CCR Updated but Operands Not Changed)	IMM DIR EXT IX2 IX1 IX SP2 SP1	A5 ii B5 dd C5 hh ll D5 ee ff E5 ff F5 9E D5 ee ff 9E E5 ff	2 3 4 4 3 3 5 4	pp rpp prpp prpp rpp rfp pprpp prpp	0	1	1	-	-	-

6.5.3.1 AND, OR, Exclusive-OR, and Complement

These instructions provide the basic AND, OR, exclusive-OR, and invert functions needed to perform Boolean logical functions.

```

        lda    #$0C          ;bit pattern 00001100
        and    #$0A          ;bit pattern 00001010
; result is.....$08.....00001000

```

```

        lda    #$35          ;bit pattern 00110101
        and    #$0F          ;bit pattern 00001111
; result is.....$05.....00000101

```

You may notice some similarity between the AND operation and the BCLR instruction. However, BCLR can be used only on memory locations \$0000–\$00FF and can clear only one bit at a time while AND can clear any combination of bits and may be used with several different addressing modes to identify the memory operand to be ANDed with A.

```

        lda    #$0C          ;bit pattern 00001100
        ora    #$0A          ;bit pattern 00001010
; result is.....$0E.....00001110

```

You may notice some similarity between the ORA operation and the BSET instruction; however, BSET can be used only on memory locations \$0000–\$00FF and can set only one bit at a time while ORA can set any combination of bits and may be used with several different addressing modes to identify the memory operand to be ORed with A.

Exclusive-OR can be used to toggle bits in an operand. One operand is considered a mask where each bit that is set in the mask corresponds to a bit value in the other operand that will be toggled (inverted). The next example reads an I/O port, exclusive-ORs it with an immediate mask value of \$03 to toggle the two least significant bits, and then writes the updated result to the I/O port.

```

402 C162 A6 0C          lda    #$0C          ;bit pattern 00001100
403 C164 A8 0A          eor    #$0A          ;bit pattern 00001010
404                      ; result is.....$06.....00000110
405
406 C166 B6 00          lda    PTAD          ;read I/O port A
407 C168 A8 03          eor    #$03          ;inverts 2 LSBs
408 C16A B7 00          sta    PTAD          ;update I/O port A

```

Complement instructions simply invert each bit of the operand. Don't confuse this with the negate instruction which performs the arithmetic operation equivalent to multiplication by minus one.

```

        lda    #$C5          ;bit pattern 11000101
        coma          ;result is  00111010

```

6.5.3.2 BIT Instruction

The BIT instruction ANDs each bit of A with the corresponding bit of the addressed memory operand (just like AND), but the result is not stored to the accumulator. The N and Z condition codes are set or cleared according to the results of the AND operation to allow conditional branches after the BIT instruction. If you load A with a mask value where each bit that is set in the mask corresponds to a bit in the memory operand to be tested, then execute a BIT instruction, the Z bit will be set if none of the tested bits were 1s.

```

lda    SCI1S1          ;read SCI status register
bit    #(mOR+mNF+mFE+mPF) ;mask of all error flags
bne    sciError         ;branch if any flags set
; A still contains undisturbed status register

sciError:                ;here if any error flags

```

6.5.4 Shift and Rotate Instructions

All of the shift and rotate instructions operate on a 9-bit field consisting of an 8-bit value in A, X, or a memory location and the C bit in the CCR. Drawings are provided in the instruction descriptions to show where the C bit fits into the shift or rotate operation. The logical shift instructions are simple shifts which shift a zero into the first bit of the value and shift the last bit into the carry bit. The arithmetic shifts treat the value to be shifted as a signed two's complement number. An arithmetic shift left is like multiplying a value by 2 and an arithmetic shift right is like dividing the number by 2. The arithmetic shift right (ASR) instruction copies the original most significant bit (MSB) back into the MSB to preserve the sign of the operand. ASL and LSL are just two different mnemonics for the same instruction because there is no functional difference between the logical and arithmetic shifts to the left.

Table 6-6. Shift and Rotate Instructions







Source Form	Operation	Address Mode	Object Code	Cycles	Cyc-by-Cyc Details	Effect on CCR	
						V I I H	I N Z C
LSL <i>opr8a</i> LSLA LSLX LSL <i>opr8,X</i> LSL ,X LSL <i>opr8,SP</i>	Logical Shift Left  (Same as ASL)	DIR INH INH IX1 IX SP1	38 dd 48 58 68 ff 78 9E 68 ff	5 1 1 5 4 6	rffwpp p p rffwpp rffw prffwpp	1 1 1 -	- 1 1 1
LSR <i>opr8a</i> LSRA LSRX LSR <i>opr8,X</i> LSR ,X LSR <i>opr8,SP</i>	Logical Shift Right 	DIR INH INH IX1 IX SP1	34 dd 44 54 64 ff 74 9E 64 ff	5 1 1 5 4 6	rffwpp p p rffwpp rffw prffwpp	1 1 1 -	- 0 1 1
ASL <i>opr8a</i> ASLA ASLX ASL <i>opr8,X</i> ASL ,X ASL <i>opr8,SP</i>	Arithmetic Shift Left  (Same as LSL)	DIR INH INH IX1 IX SP1	38 dd 48 58 68 ff 78 9E 68 ff	5 1 1 5 4 6	rffwpp p p rffwpp rffw prffwpp	1 1 1 -	- 1 1 1

Table 6-6. Shift and Rotate Instructions (Continued)

Source Form	Operation	Address Mode	Object Code	Cycles	Cyc-by-Cyc Details	Effect on CCR	
						V I 1 H	IN Z C
ASR <i>opr8a</i> ASRA ASRX ASR <i>opr8,X</i> ASR ,X ASR <i>opr8,SP</i>	Arithmetic Shift Right 	DIR INH INH IX1 IX SP1	37 dd 47 57 67 ff 77 9E 67 ff	5 1 1 5 4 6	rfwpp p p rfwpp rfwp prfwpp	† 1 1 -	- † † †
ROL <i>opr8a</i> ROLA ROLX ROL <i>opr8,X</i> ROL ,X ROL <i>opr8,SP</i>	Rotate Left through Carry 	DIR INH INH IX1 IX SP1	39 dd 49 59 69 ff 79 9E 69 ff	5 1 1 5 4 6	rfwpp p p rfwpp rfwp prfwpp	† 1 1 -	- † † †
ROR <i>opr8a</i> RORA RORX ROR <i>opr8,X</i> ROR ,X ROR <i>opr8,SP</i>	Rotate Right through Carry 	DIR INH INH IX1 IX SP1	36 dd 46 56 66 ff 76 9E 66 ff	5 1 1 5 4 6	rfwpp p p rfwpp rfwp prfwpp	† 1 1 -	- † † †

Including the carry bit in the shifts and rotates allows extension of these operations to multibyte values. The following examples show a 24-bit value being shifted either right or left.

```

; 24-bit left shift
        clc                ;clear C bit
; initial condition sum24 = hhhh hhhh : mmmm mmmm : 1111 1111 : 0
        lsl    sum24+2     ;C to LSB of low byte
; now sum24 = hhhh hhhh : mmmm mmmm : C=1(7) : 1111 1110
        rol    sum24+1     ;rotate middle byte
; now sum24 = hhhh hhhh : C=m(7) : mmmm mmm1 : 1111 1110
        rol    sum24       ;rotate high byte
; now sum24 = C=h(7) : hhhh hhhm : mmmm mmm1 : 1111 1110

```

```

; 24-bit right shift
        clc                ;clear C bit
; initial condition sum24 = 0 : hhhh hhhh : mmmm mmmm : 1111 1111
        lsr    sum24       ;C to MSB of high byte
; now sum24 = 0hhh hhhh : C=h(0) : mmmm mmmm : 1111 1111
        rol    sum24+1     ;rotate middle byte
; now sum24 = 0hhh hhhh : hmmm mmmm : C=m(0) : 1111 1110
        rol    sum24+2     ;rotate low byte
; now sum24 = 0hhh hhhm : hmmm mmmm : m111 1111 : C=1(0)

```

Figure 6-5. Multibyte Shifts

6.5.5 Jump, Branch, and Loop Control Instructions

The instructions in this group cause a change of flow which means that the CPU loads a new address into the program counter so program execution continues at a location other than the next memory location after the current instruction.

Jump instructions cause an unconditional change in the execution sequence to a new location in a program. Branch and loop control instructions cause a conditional change in the execution sequence. Branch and loop control instructions use relative addressing mode to conditionally branch to a location that is relative to the location of the branch. Processor status indicators in the CCR control whether a conditional branch or loop control instruction will branch to a new address or simply continue to the next instruction in the program. BRA is a special case because the branch *a/ways* occurs and BRN is special because the branch is *never* taken (this is functionally equivalent to a 2-byte, 3-cycle NOP). BIL and BIH are special because they use the state of the IRQ pin rather than the condition of a bit(s) in the CCR to decide whether to branch.

Table 6-7. Jump and Branch Instructions

Source Form	Operation	Address Mode	Object Code	Cycles	Cyc-by-Cyc Details	Effect on CCR	
						V 1 1 H	I N Z C
JMP <i>opr8a</i> JMP <i>opr16a</i> JMP <i>opr16,X</i> JMP <i>opr8,X</i> JMP <i>,X</i>	Jump PC ← Jump Address	DIR EXT IX2 IX1 IX	BC dd CC hh ll DC ee ff EC ff FC	3 4 4 3 3	ppp pppp pppp ppp ppp	- 1 1 -	- - - -
BRA <i>rel</i>	Branch Always (if I = 1)	REL	20 rr	3	ppp	- 1 1 -	- - - -
BRN <i>rel</i>	Branch Never (if I = 0)	REL	21 rr	3	ppp	- 1 1 -	- - - -
BEQ <i>rel</i>	Branch if Equal (if Z = 1)	REL	27 rr	3	ppp	- 1 1 -	- - - -
BNE <i>rel</i>	Branch if Not Equal (if Z = 0)	REL	26 rr	3	ppp	- 1 1 -	- - - -
BCC <i>rel</i>	Branch if Carry Bit Clear (if C = 0)	REL	24 rr	3	ppp	- 1 1 -	- - - -
BCS <i>rel</i>	Branch if Carry Bit Set (if C = 1) (Same as BLO)	REL	25 rr	3	ppp	- 1 1 -	- - - -
BPL <i>rel</i>	Branch if Plus (if N = 0)	REL	2A rr	3	ppp	- 1 1 -	- - - -
BMI <i>rel</i>	Branch if Minus (if N = 1)	REL	2B rr	3	ppp	- 1 1 -	- - - -
BIL <i>rel</i>	Branch if IRQ Pin Low (if IRQ pin = 0)	REL	2E rr	3	ppp	- 1 1 -	- - - -
BIH <i>rel</i>	Branch if IRQ Pin High (if IRQ pin = 1)	REL	2F rr	3	ppp	- 1 1 -	- - - -
BMC <i>rel</i>	Branch if Interrupt Mask Clear (if I = 0)	REL	2C rr	3	ppp	- 1 1 -	- - - -
BMS <i>rel</i>	Branch if Interrupt Mask Set (if I = 1)	REL	2D rr	3	ppp	- 1 1 -	- - - -
BHCC <i>rel</i>	Branch if Half Carry Bit Clear (if H = 0)	REL	28 rr	3	ppp	- 1 1 -	- - - -
BHCS <i>rel</i>	Branch if Half Carry Bit Set (if H = 1)	REL	29 rr	3	ppp	- 1 1 -	- - - -
BLT <i>rel</i>	Branch if Less Than (if N ⊕ V = 1) (Signed)	REL	91 rr	3	ppp	- 1 1 -	- - - -
BLE <i>rel</i>	Branch if Less Than or Equal To (if Z I (N ⊕ V) = 1) (Signed)	REL	93 rr	3	ppp	- 1 1 -	- - - -
BGE <i>rel</i>	Branch if Greater Than or Equal To (if N ⊕ V = 0) (Signed)	REL	90 rr	3	ppp	- 1 1 -	- - - -

Table 6-7. Jump and Branch Instructions (Continued)

Source Form	Operation	Address Mode	Object Code	Cycles	Cyc-by-Cyc Details	Effect on CCR	
						V 1 1 H	I N Z C
BGT <i>rel</i>	Branch if Greater Than (if $Z \mid (N \oplus V) = 0$) (Signed)	REL	92 <i>rr</i>	3	ppp	- 1 1 -	- - - -
BLO <i>rel</i>	Branch if Lower (if $C = 1$) (Same as BCS)	REL	25 <i>rr</i>	3	ppp	- 1 1 -	- - - -
BLS <i>rel</i>	Branch if Lower or Same (if $C \mid Z = 1$)	REL	23 <i>rr</i>	3	ppp	- 1 1 -	- - - -
BHS <i>rel</i>	Branch if Higher or Same (if $C = 0$) (Same as BCC)	REL	24 <i>rr</i>	3	ppp	- 1 1 -	- - - -
BHI <i>rel</i>	Branch if Higher (if $C \mid Z = 0$)	REL	22 <i>rr</i>	3	ppp	- 1 1 -	- - - -

6.5.5.1 Unconditional Jump and Branch

Jump (JMP), branch always (BRA), and branch never (BRN) are unconditional and do not depend on the state of any CCR bits. Jump may be used to go to any memory location in the 64-Kbyte address space while branch instructions are limited to destinations within -128 to +127 locations from the address immediately after the branch offset byte.

The following example illustrates the use of a JMP instruction to extend the range of a conditional branch. For every conditional branch instruction there is another branch that uses the opposite condition. For example the opposite of a branch if equal (BEQ) instruction is the branch if not equal (BNE) instruction. Suppose you wrote the instruction:

```
;          beq    farAway    ;more than 128 locs away
```

and the assembler flagged an error because farAway was more than 128 locations away. You can replace the BEQ with a BNE that branches around a jump instruction like this:

```
          bne    aroundJ    ;skip if NOT equal
          jmp     farAway    ;jump if equal
aroundJ:                ;here if not equal
```

6.5.5.2 Simple Branches

The simple branches only depend on the state of a single condition (a CCR bit or the IRQ pin state).

Table 6-8. Simple Branch Summary

Branch Condition	Branch if True	Branch if False
Z	BEQ	BNE
C	BCS	BCC
N	BMI	BPL
IRQ pin	BIH	BIL
I	BMS	BMC
H	BHCS	BHCC